

Using SAS® to Calculate Arizona Judicial Salaries

Mark McDermott and Humberto Cisneros, Arizona Supreme Court, Phoenix, AZ

ABSTRACT

Justices of the Peace in Arizona are judges in courts of limited jurisdiction. These courts hear a wide variety of cases. The case types these judges adjudicate vary widely, from felonies to civil traffic violations, such as speeding. Based in Arizona law, justices of the peace are compensated using a complicated mathematical formula, along with pay categories based on a percentage of a superior court judge's salary. The formula, based on the number of cases filed into each justice court, is then translated into what is called "judicial productivity credits". It is these credits that are used to determine the pay level of each justice of the peace. Research and Statistics staff of the Supreme Court are mandated by law to provide data used to calculate judicial productivity credits and utilize SAS®, not only to determine the case filings in each court, but SAS® programs actually calculate and generate judicial productivity credits.

AN OVERVIEW OF JUDICIAL SALARIES IN ARIZONA

Judges in Arizona are compensated by two mutually exclusive methods. Either their salary is established by state statutes (laws), or their salary is set or negotiated by the city or town hiring them. The majority of judges in Arizona courts are paid based on salaries established by state statutes. This is true for all appellate court judges, superior court judges, and judges known as a justice of the peace. Magistrates, or judges working for a city or town court, are compensated by the city or town hiring them. These salaries can be established by the municipality, or they can be negotiated by the prospective applicant for the judicial position.

For the purposes of this paper, we will examine how a justice of the peace is compensated. Of the judicial positions compensated by state statutes, only a justice of the peace is compensated using a rather complex, yet cumbersome, mathematical formula. The other judicial positions compensated using state statutes are paid a set salary. For example, the chief justice of the Arizona Supreme Court is paid a set salary of \$129,150 each year while a superior court judge is paid \$120,750. In contrast, a justice of the peace is compensated based on an average of the number of cases filed into his or her court every two calendar years. These case filings are then plugged into a mathematical formula to generate what is known as judicial productivity credits.

HOW ARE CASE FILINGS MONITORED

In accordance with state statute, it is the Arizona Supreme Court's responsibility to collect and compile statistics for every court in the state (excluding tribal courts). The Supreme Court compiles case activity, such as case filings and terminations, on a monthly basis via a prescribed statistical report each court must file on a monthly basis. The Supreme Court also compiles revenue data using a quarterly survey, as well as a yearly expenditure survey and personnel survey each court and probation department must submit. These data are entered into SAS® databases, and the fiscal year Data Report is then published from these databases. A portion of this data is then used to calculate judicial productivity credits which are based on a two year average of case filings in the justice courts.

COMPENSATING A JUSTICE OF THE PEACE

As stated in the introduction, judges in Arizona known as a justice of the peace are compensated using a rather complex mathematical formula found in state statute (Arizona Revised Statutes § 22-125). This statute can be found in the reference section of this report. Basically, the number of charges filed into the court are summed according to their filing type (felonies, misdemeanors, civil traffic, criminal traffic and civil cases), divided by ten (except for civil traffic charges and other criminal traffic, which are divided by sixty), and the results added together to generate judicial productivity credits. In accordance with the statute, judicial productivity credits are then categorized, and each category is associated with a set level of compensation which is a percentage of a superior judge's salary. The statute specifies seven levels of compensation for justices of the peace, from a low of \$30, 187.50 (25% of a superior court judge's salary) to a high of \$84,525.00 (70% of a superior court judge's salary).

MACROS FOR SETTING DATA AND SELECTING THE VARIABLES

The following SAS® code uses various macros to select the desired justice courts by county and the desired years needed to produce the judicial productivity credits. Other macros are used to sort and print report results. Using the new features found in SAS® version 8 allowed us to use more descriptive names for the many variables required in the SAS® program.

```
OPTIONS LS=98 PS=58 FMTSEARCH=(COURTS) NODATE NONUMBER SYSPRINT='\\SUPREME_3\PRTcs05';  
FILENAME OUT '\\SUPREME_3\PRTcs05';
```

```
MACRO PRINTT PROC PRINT NOOBS UNIFORM SPLIT = '*' %
```

```
MACRO SORTBY      PROC SORT; BY %
MACRO SORTT      PROC SORT DATA = %
```

```
/* *****
* ESTIMATED PRODUCTIVITY CREDITS USING 2004-2005 DATA--INDIVIDUAL COURTS BY COUNTY *
* (SAS/AF: 5 - OTHER REPORTS *
* 15 - JP PRODUCTIVITY CREDIT REPORT (1ST YR = 2004, 2ND YR= 2005) *
***** */
```

```
%LET Selcnty = 02; * Select County to be Processed;
%LET SelYear1 = 2004; * Select first year of case filings;
%LET SelYear2 = 2005; * Select Second year of case filings;
```

EXTRACTING AND SUMMARIZING CASE DATA

The following SQL modules extract and summarize variables in the justice court case filing data. The first module extracts data for the first year selected for productivity credit analysis, while the second module handles the same procedure for the second year selected.

```
PROC SQL;
CREATE TABLE Y1Data AS
SELECT Courtid, Count(*) AS Months_count, County, Month,
SUM(F_TOTCC, TI_TOTCC, HEAR_JUV, F_PB, F_FUG, F_DV, F_HAR, -OFFSCTRL) as Civil_Filings,
SUM(F_FEL, TI_FEL) as Felony_Filings,
SUM(F_TOTMS, TI_TOTMS, F_FTATR, TI_FTATR) as
Misdemeanor_Filings,
SUM(F_DUI, F_SER, TI_DUI, TI_SER, TF_EXSPD) as
Criminal_Traffic_Filings,
SUM(F_OTH, TI_OTH, F_TRFCV, TI_TRFCV, -TF_EXSPD) as
Civil_Traffic_Filings
FROM COURTS.JPDATA
WHERE YEAR4 = &selyear1 and county = &selcnty
GROUP by Courtid
Order by Courtid;
QUIT;
```

```
PROC SQL;
CREATE TABLE Y2Data AS
SELECT Courtid, Count(*) AS Months_count, County, Month,
SUM(F_TOTCC, TI_TOTCC, HEAR_JUV, F_PB, F_FUG, F_DV, F_HAR, -OFFSCTRL) as Civil_Filings,
SUM(F_FEL, TI_FEL) as Felony_Filings,
SUM(F_TOTMS, TI_TOTMS, F_FTATR, TI_FTATR) as
Misdemeanor_Filings,
SUM(F_DUI, F_SER, TI_DUI, TI_SER, TF_EXSPD) as
Criminal_Traffic_Filings,
SUM(F_OTH, TI_OTH, F_TRFCV, TI_TRFCV, -TF_EXSPD) as
Civil_Traffic_Filings
FROM COURTS.JPDATA
WHERE YEAR4 = &selyear2 and county = &selcnty
GROUP by Courtid
Order by Courtid;
QUIT;
```

PROC MEANS

This procedure is used to summarize case filings by case category, such as misdemeanors and felonies. The summarized data is output to a separate SAS® database.

```
PROC MEANS DATA=Y1Data SUM NOPRINT;
BY COUNTY COURTTID;
VAR Civil_Filings
Felony_Filings
Misdemeanor_Filings
Criminal_Traffic_Filings
Civil_Traffic_Filings;
OUTPUT
OUT=Y1SUM
SUM=Y1_Civil_Filings
Y1_Felony_Filings
```

```

Y1_Misdemeanor_Filings
Y1_Criminal_Traffic_Filings
Y1_Civil_Traffic_Filings;

```

```

PROC MEANS DATA=Y2Data SUM NOPRINT;
  BY COUNTY COURTID;
  VAR Civil_Filings
      Felony_Filings
      Misdemeanor_Filings
      Criminal_Traffic_Filings
      Civil_Traffic_Filings;
  OUTPUT
  OUT=Y2SUM
  SUM=Y2_Civil_Filings
      Y2_Felony_Filings
      Y2_Misdemeanor_Filings
      Y2_Criminal_Traffic_Filings
      Y2_Civil_Traffic_Filings;

```

DATA STEP TO ESTIMATE ANY MISSING DATA

In this step, we use an array with five elements to extrapolate missing monthly data. The procedure also counts the number of missing months of data. This enables us to account for the amount of data that is available for extrapolation. Estimates are used for planning purposes only.

```

DATA Y1PROJ;
  SET Y1SUM;
  ARRAY SUMDATA(5)
    Y1_Civil_Filings
    Y1_Felony_Filings
    Y1_Misdemeanor_Filings
    Y1_Criminal_Traffic_Filings
    Y1_Civil_Traffic_Filings;
  DO I = 1 TO 5;
    SUMDATA{I} = (SUMDATA{I}/_FREQ_)*12;
  END;
  Year1_Number_of_Months = _freq_;
  if Year1_Number_of_Months = . then Year1_Number_of_Months = 0;

DATA Y2PROJ;
  SET Y2SUM;
  ARRAY SUMDATA(5)
    Y2_Civil_Filings
    Y2_Felony_Filings
    Y2_Misdemeanor_Filings
    Y2_Criminal_Traffic_Filings
    Y2_Civil_Traffic_Filings;
  DO I = 1 TO 5;
    SUMDATA{I} = (SUMDATA{I}/_FREQ_)*12;
  END;
  Year2_Number_of_Months = _freq_;
  if Year2_Number_of_Months = . then Year2_Number_of_Months = 0;

```

MERGING TWO YEARS OF MONTHLY DATA

This step merges year one case filing data and year two case filing data to create a two year average. This step also replaces year two data with year one data when all second year months are missing.

```

DATA JPCDATA;
  MERGE Y1PROJ Y2PROJ;
  BY COUNTY COURTID;
  IF Y2_Civil_Filings = . THEN Y2_Civil_Filings = Y1_Civil_Filings;
  IF Y2_Felony_Filings = . THEN Y2_Felony_Filings = Y1_Felony_Filings;
  IF Y2_Misdemeanor_Filings = . THEN Y2_Misdemeanor_Filings =
Y1_Misdemeanor_Filings;
  IF Y2_Criminal_Traffic_Filings = . THEN Y2_Criminal_Traffic_Filings =
Y1_Criminal_Traffic_Filings;
  IF Y2_Civil_Traffic_Filings = . THEN Y2_Civil_Traffic_Filings =
Y1_Civil_Traffic_Filings;
  Total_Year1=SUM(Y1_Civil_Filings,Y1_Felony_Filings,Y1_Misdemeanor_Filings,

```

```

                Y1_Criminal_Traffic_Filings,Y1_Civil_Traffic_Filings);
Total_Year2=SUM(Y2_Civil_Filings,Y2_Felony_Filings,Y2_Misdemeanor_Filings,
                Y2_Criminal_Traffic_Filings,Y2_Civil_Traffic_Filings);
Avg_Civil_Filings      = Mean(Y1_Civil_Filings,Y2_Civil_Filings);
Avg_Felony_Filings     = Mean(Y1_Felony_Filings,Y2_Felony_Filings);
Avg_Misdemeanor_Filings =
Mean(Y1_Misdemeanor_Filings,Y2_Misdemeanor_Filings);
Avg_Criminal_Traffic_Filings =
Mean(Y1_Criminal_Traffic_Filings,Y2_Criminal_Traffic_Filings);
Avg_Civil_Traffic_Filings =
Mean(Y1_Civil_Traffic_Filings,Y2_Civil_Traffic_Filings);

    JPC_Civil_Filings      = Avg_Civil_Filings/10;
    JPC_Felony_Filings     = Avg_Felony_Filings/10;
    JPC_Misdemeanor_Filings = Avg_Misdemeanor_Filings/10;
    JPC_Criminal_Traffic_Filings = Avg_Criminal_Traffic_Filings/10;
    JPC_Civil_Traffic_Filings = Avg_Civil_Traffic_Filings/60;
    Total_JPC=SUM(JPC_Civil_Filings,JPC_Felony_Filings,JPC_Misdemeanor_Filings,
                  JPC_Criminal_Traffic_Filings,JPC_Civil_Traffic_Filings);
    JPTXT=TRIM(PUT(COURTID,JPCTCAPS.));
    CNTYTXT=TRIM(PUT(COUNTY,CNTYCAP.))||' COUNTY';
    YEAR1 = &selyear1;
    YEAR2 = &selyear2;
SORTBY COUNTY JPTXT;

```

REPORT OUTPUT

Data null step was used to produce desired report based on the various elements needed to produce judicial productivity credits.

```

DATA _NULL_;
FILE '\\SUPREME_3\PRTcs05';
PUT '1B266B3453'X; /* ELITE PITCH (12.0) */
title;
DATA _NULL_;
SET JPCDATA END=ENDFILE;
BY COUNTY JPTXT;
*FILE OUT PRINT N=1 NOTITLES;
FILE PRINT;
IF FIRST.COUNTY THEN X=0;
IF MOD(X,7)=0.0 THEN DO;
PUT _PAGE_
   @27 'JUSTICE OF THE PEACE PRODUCTIVITY CREDITS' /
   @13 'BASED ON ACTUAL OR ESTIMATED FILINGS FOR CALENDAR YEARS' +1 Year1 'AND' +1
Year2 ///
   @1 CNTYTXT @;
IF X THEN PUT '(CONTINUED)'; ELSE PUT;
PUT @47 'ACTUAL OR ESTIMATED FILINGS*' OVERPRINT @36 49*'_ ' /
   @56 'MISDE-' @65 'SERIOUS' @77 'OTHER' @94 '# OF' /
   @1 'COURT' @37 'CIVIL' @46 'FELONY' @56 'MEANOR'
   @68 'TRAF' @78 'TRAF' @87 'TOTAL' @94 'MTHS' OVERPRINT @1 98*'_ ' /;
END;
PUT @1 COURTID JPCTCAPS.
   @26 Year1 +0 ':'
   @36 Y1_Civil_Filings COMMA6.
   @46 Y1_Felony_Filings COMMA6.
   @56 Y1_Misdemeanor_Filings COMMA6.
   @66 Y1_Criminal_Traffic_Filings COMMA6.
   @76 Y1_Civil_Traffic_Filings COMMA6.
   @86 Total_Year1 COMMA6.
   @93 '[' @94 Year1_Number_of_Months 2. @97 ']' /
@26 Year2 +0 ':'
   @36 Y2_Civil_Filings COMMA6.
   @46 Y2_Felony_Filings COMMA6.
   @56 Y2_Misdemeanor_Filings COMMA6.
   @66 Y2_Criminal_Traffic_Filings COMMA6.
   @76 Y2_Civil_Traffic_Filings COMMA6.
   @86 Total_Year2 COMMA6.
   @93 '[' @94 Year2_Number_of_Months 2. @97 ']' /

```

```

@24 'Average:'
    @36 AVG_Civil_Filings COMMA6.2
    @46 AVG_Felony_Filings COMMA6.2
    @56 AVG_Misdemeanor_Filings COMMA6.2
    @66 AVG_Criminal_Traffic_Filings COMMA6.2
    @76 AVG_Civil_Traffic_Filings COMMA6.2 /
    @21 'Divided By:'
    @40 '10' @50 '10' @60 '10' @70 '10' @80 '60' OVERPRINT @23 62*'_ '
@86 11*'_ ' /
@10 'Estimated JPC Credits:'
    @36 JPC_Civil_Filings COMMA6.2
    @46 JPC_Felony_Filings COMMA6.2
    @56 JPC_Misdemeanor_Filings COMMA6.2
    @66 JPC_Criminal_Traffic_Filings COMMA6.2
    @76 JPC_Civil_Traffic_Filings COMMA6.2
    @86 '| ' @87 Total_JPC COMMA8.2 @97 '| ' OVERPRINT @23 62*'_ ' + 2 @86 11*'_ ' //;
IF COURTID=1001 THEN Total_JPC=Total_JPC/6; /*CALCULATION FOR PIMA CONSOLIDATED
JUSTICE COURT*/
IF COURTID=1001 THEN PUT @23 'PRODUCTIVITY CREDITS PER JUDGE'
    @87 Total_JPC COMMA8.2 OVERPRINT @23 74*'_ ' ;
IF COURTID=0801 THEN Total_JPC=Total_JPC/2; /*CALCULATION FOR KINGMAN-CERBAT JUSTICE
COURT*/
IF COURTID=0801 THEN PUT @23 'PRODUCTIVITY CREDITS PER JUDGE'
    @87 Total_JPC COMMA8.2 OVERPRINT @23 74*'_ ' ;
IF LAST.COUNTY THEN
    PUT @1 98*'_ ' //
    @1 '* IF "NO OF MONTHS" (RIGHTMOST COLUMN, IN BRACKETS) EQUALS "12", THE TOTAL
FILINGS ARE' /
    @4 '"ACTUAL". OTHERWISE, THEY ARE "ESTIMATED" BASED ON EXTRAPOLATING THE
FILINGS FOR THE MONTHS' /
    @4 'OF ACTUAL DATA ENTERED (IN BRACKETS) THROUGH THE END OF THE YEAR. IF NO
DATA EXISTS FOR' /
    @4 'THE SECOND YEAR (I.E., "NO OF MONTHS" IN BRACKETS IS ZERO), THEN FIGURES
FROM THE FIRST YEAR' /
    @4 'WERE SUBSTITUTED. "FILINGS" INCLUDE CASES "TRANSFERRED IN" TO THE COURT.' ;
X+1;
RETAIN X;

DATA _NULL_;
FILE '\\SUPREME_3\PRTcs05';
PUT '1B45'X; /* RESET */

RUN;

```

REPORT RESULTS

The following is a sample report based on the results of the SAS® program.

JUSTICE OF THE PEACE PRODUCTIVITY CREDITS
BASED ON ACTUAL OR ESTIMATED FILINGS FOR CALENDAR YEARS 2004 AND 2005

COCHISE COUNTY

ACTUAL OR ESTIMATED FILINGS*

COURT	CIVIL	FELONY	MISDE- MEANOR	SERIOUS TRAF	OTHER TRAF	TOTAL	# OF
							MTHS
<p>2004 :</p>	333	182	964	1,366	4,042	6,887	[12]

2005 :	340	180	1,108	1,552	5,444	8,624 [3]
Average:	336.50	181.00	1036.0	1459.0	4743.0	
Divided By:	10	10	10	10	60	
<hr/>						
Estimated JPC Credits:	33.65	18.10	103.60	145.90	79.05	380.30
<hr/>						

CONCLUSION

In Arizona, justices of the peace comprise roughly 24 percent of all full-time judges in the state, and these 86 judges are the only judges compensated using a rather complex mathematical formula based on the number of charges filed into each of their respective courts. According to state law, the Arizona Supreme Court must produce data used to pay the salaries for all justices of the peace in the state. SAS® is a valuable tool to accomplish this rather complex, but important task.

REFERENCES

(ARIZONA REVISED STATUTE § 22-125)

22-125. Justice of the peace compensation; judicial productivity credits

- A. The annual salary of each justice of the peace is determinable on the basis of the total judicial productivity credits of each court as reported in statistics compiled by the Arizona supreme court according to the following formula:
 1. Total civil filings, excluding those small claims filings heard by a volunteer hearing officer, divided by ten equals _____ judicial productivity credits. Juvenile hearings pursuant to section 8-323, excluding juvenile traffic violations, shall be counted as civil filings.
 2. Total felony complaint filings divided by ten equals _____ judicial productivity credits.
 3. Total misdemeanor filings divided by ten equals _____ judicial productivity credits.
 4. Total traffic filings of a violation of sections 28-662 through 28-664, 28-693, 28-702.01, 28-702.04, 28-708, 28-1381 and 28-1382 divided by ten equals _____ judicial productivity credits.
 5. Total of all other traffic filings divided by sixty equals _____ judicial productivity credits.
- B. Total judicial productivity credits for a justice of the peace is the sum of the figures computed in subsection A of this section.
- C. The board of supervisors shall use the average number of cases filed annually in each justice court for the two year period ending December 31, 1979 to determine the initial judicial productivity credits for each of the case load categories described in subsection A of this section. The board of supervisors shall compute judicial productivity credits every two years thereafter using the average annual filings for the preceding two year period.
- D. In determining the compensation levels appropriate to the case load of each justice of the peace, the board of supervisors shall use the total judicial productivity credits as follows:
 1. Five hundred or more judicial productivity credits, seventy per cent of the annual compensation payable to a judge of the superior court.
 2. Two hundred to four hundred ninety-nine total judicial productivity credits, sixty-five per cent of the annual compensation payable to a judge of the superior court.
 3. One hundred fifty to one hundred ninety-nine total judicial productivity credits, fifty-five per cent of the annual compensation payable to a judge of the superior court.
 4. One hundred to one hundred forty-nine total judicial productivity credits, fifty per cent of the annual compensation payable to a judge of the superior court.
 5. Fifty to ninety-nine total judicial productivity credits, forty-five per cent of the annual compensation payable to a judge of the superior court.
 6. Twenty-five to forty-nine total judicial productivity credits, thirty-five per cent of the annual compensation payable to a judge of the superior court.
 7. Twenty-four or fewer total judicial productivity credits, a part-time case load for one justice of the peace, twenty-five per cent of the annual compensation payable to a judge of the superior court.
- E. If a justice court is not assigned clerical help, the board of supervisors shall multiply the total judicial productivity credits by two for purposes of determining compensation.
- F. If the board of supervisors divides a justice precinct into two or more precincts, the board shall set the level of compensation consistent with the intent of this section until such time as statistics are available for computation of compensation levels using the formula prescribed by this section.

G. The board of supervisors shall review and adjust the level of compensation for each justice of the peace within the county every two years beginning January 1, 1983.

H. The judicial productivity credits for any justice court precinct shall not exceed twelve hundred credits. If the total judicial productivity credits of a justice court precinct exceed twelve hundred credits, the county board of supervisors shall create sufficient courts, or redraw the justice court precinct boundaries according to section 22-101, in order to reduce the judicial productivity credits for any precinct which exceeds that limit.

(TABLE LISTING CURRENT JUDICIAL PRODUCTIVITY CREDITS AND SALARIES)

**ARIZONA SUPREME COURT
ADMINISTRATIVE OFFICE OF THE COURTS
COURT SERVICES DIVISION**

JUSTICE OF THE PEACE PRODUCTIVITY CREDITS

The following table lists the productivity credits of the office of Justice of the Peace and the corresponding salaries in accordance with A.R.S. 22-125:

COUNTY	COURT	PRODUCTIVITY CREDITS	SALARY EFFECTIVE 2005-2006
Apache	Chinle	39.51	42,262.50
	Puerco	202.14	78,487.50
	Round Valley	144.31	60,375
	St. Johns	63.23	54,337.50
Cochise	Benson	341.41	78,487.50
	Bisbee	265.30	78,487.50
	Bowie	287.58	78,487.50
	Douglas	530.29	84,525
	Sierra Vista	918.25	84,525
	Willcox	283.57	78,487.50
Coconino	Flagstaff	1,192.51	84,525
	Fredonia	58.64	54,337.50
	Page	187.06	66,412.50
	Williams	261.28	78,487.50
Gila	Globe Regional	400.88	78,487.50
	Payson	483.48	78,487.50
Graham	Pima	82.64	54,337.50
	Safford	259.70	78,487.50
Greenlee	Clifton	55.62	54,337.50
	Duncan	69.92	54,337.50
La Paz	Parker	327.83	78,487.50

	Salome	136.65	60,375
	Quartzsite	307.53	78,487.50
Maricopa	Buckeye	761.35	84,525
	Central Phoenix	1,085.97	84,525
	Chandler	1,132.35	84,525
	East Mesa	1,095.33	84,525
	East Phoenix #1	1,194.42	84,525
	East Phoenix #2	1,182.18	84,525
	East Tempe	1,011.38	84,525
	Gila Bend	190.42	66,412.50
	Glendale	1,148.13	84,525
	Maryvale	858.65	84,525
	North Mesa	734.59	84,525
	North Valley	1,264.08	84,525
	Northeast Phoenix	1,179.83	84,525
	Northwest Phoenix	1,006.94	84,525
	Peoria	1,171.16	84,525
	Scottsdale	1,304.51	84,525
	South Mesa/Gilbert	979.78	84,525
	South Phoenix	734.11	84,525
	Tolleson	1,057.79	84,525
	West Mesa	1,234.76	84,525
	West Phoenix	1,043.70	84,525
	West Tempe	974.14	84,525
	Wickenburg	273.75	78,487.50
Mohave	Bullhead City	623.23	84,525
	Colorado (Moccasin)	175.60	66,412.50
	Kingman/Cerbat	703.19	169,050
	Lake Havasu	389.04	78,487.50
Navajo	Holbrook	300.16	78,487.50
	Kayenta	11.72	30,187.50
	Show Low	168.57	66,412.50

	Snowflake	354.89	78,487.50
	Pinetop/Lakeside	202.78	78,487.50
	Winslow	374.23	78,487.50
Pima	Ajo	286.29	78,487.50
	Green Valley	260.48	78,487.50
	Pima Consolidated	1,695.47	507,150
Pinal	Apache Junction	462.15	78,487.50
	Casa Grande	752.25	84,525
	Eloy	295.77	78,487.50
	Florence	205.40	78,487.50
	Mammoth	122.48	60,375
	Maricopa	184.90	66,412.50
	Oracle	206.50	78,487.50
	Superior	191.13	66,412.50
Santa Cruz	East Santa Cruz	70.98	54,337.50
	Nogales	456.10	78,487.50
Yavapai	Bagdad-Yarnell	240.42	78,487.50
	Mayer	404.38	78,487.50
	Prescott	823.57	84,525
	Seligman	153.15	66,412.50
	Verde Valley	569.85	84,525
Yuma	Somerton	139.61	60,375
	Wellton	154.01	66,412.50
	Yuma	1,030.83	84,525
TOTAL			\$6,339,375.00

Pima County Consolidated Justice Court (Pima County) consists of 6 justice of the peace precincts.
Kingman/Cerbat Justice Court (Mohave County) consists of 2 justice of the peace precincts.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Mark J. McDermott
Arizona Supreme Court, Administrative Office of the Courts
1501 W Washington Street, Suite 410
Phoenix, AZ 85007
Work Phone: (602) 542-9572
Fax: (602) 542-9480
Email: mmcdermott@courts.az.gov
Web: www.supreme.state.az.us/stats

Humberto Cisneros
Arizona Supreme Court, Administrative Office of the Courts
1501 W Washington Street, Suite 410
Phoenix, AZ 85007
Work Phone: (602) 542-9159
Fax: (602) 542-9480
Email: hcisneros@courts.az.gov
Web: www.supreme.state.az.us/stats

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.